/*
Patrick Ishizuka
Proposal title: Preferences or Constraints? Education, Gender, and Parenting Attitudes



HYPOTHESES

Stated-Hyp1: "For open-ended vignettes, attitudes should reflect theorized education-based preferences."
	
	- Notes: author means open-ended outcome measures. Cannot test this because analysis requires structural topic modeling. See notes at the end. 


Stated-Hyp2: "Respondents should advise concerted cultivation parenting for mothers, but not for fathers" (p.3).
	
	- Note: this hypothesis follows stated hypothesis 1, and pertains to the open-ended questions. Cannot test this.

	
Stated-Hyp2: "For closed-ended vignettes, high-education respondents should rate concerted cultivation most highly, while low-education respondents should rate natural growth most highly." (p. 3)

	Test-Hyp1: Within concerted cultivation vignette, rating for this parenting style will be higher among high-education vs. low-education respondents.

Stated-Hyp3: "mothers should be rated more negatively than fathers for both parenting styles" (p. 4)

	Test-Hyp2: Within concerted cultivation vignette, rating for parenting style will be lower for mothers vs. fathers.
	
	Test-Hyp3: Within natural growth vignette, rating for parenting style will be lower for mothers vs. fathers.

Stated-Hyp4: "For open-ended vignettes, both low- and high education
respondents should advise concerted cultivation parenting, and respondents should advise concerted cultivation for mothers and faters." (p. 4) 
	
	- Note: open-ended response. Cannot test. See notes.

Stated-Hyp4: for closed-ended vignettes, both low- and high-education respondents should rate concerted cultivation most highly, and there should be no differences in how mothers and fathers are rated for either parenting approach. (p. 4)
	
	Test-Hyp4: Within low-education respondents, rating will be higher for concerted cultivation vs. natural growth parenting.
	
	Test-Hyp5: Within high-education respondents, rating will be higher for concerted cultivation vs. natural growth parenting.
	
********************************************************************************
NOTES

- Analysis plan states author will use ologit. We are using linear regressions 
for consistency across studies.

- there are 2 open-ended questions. the analysis plan for these entails 
structured topic modeling. This coding will inevitably be subjective. We are 
not pursuing these because this is not a clear-cut hypothesis test.

*/

clear all
use "ishizuka661.dta", clear

********************************************************************************

* MEASURES OF EXPERIMENTAL MANIPULATION

* parenting style: concerted cultivation vs. natural growth
	egen parentingtype=rowtotal(dov_momvignette2_1 dov_momvignette2_2 dov_momvignette2_3 dov_momvignette2_4 dov_dadvignette2_1 dov_dadvignette2_2 dov_dadvignette2_3 dov_dadvignette2_4)

	tab parentingtype

	recode parentingtype (1 2 5 6 9 10 13 14 17 18 21 22=1) (3 4 7 8 11 12 15 16 19 20 23 24=0), gen(conc_cult)
	tab conc_cult, mis

	lab def concult 1 "concered cultivation" 0 "natural growth"
	lab val conc_cult concult
	tab conc_cult, mis

* parent gender
	tab dov_q2q3_parent
	recode dov_q2q3_parent (1=1) (2=0), gen(parent_mother)
	tab parent_mother, mis


* OUTCOME
	* rating of vignette's parenting
	tab q2 // reverse coded
	recode q2 (6=1) (5=2) (4=3) (3=4) (2=5) (1=6) (-1=.r), gen(rating)
	lab def rating 1 "poor" 2 "not very good" 3 "ok" 4 "good" 5 "very good" 6 "excellent" .r "refused"
	lab val rating rating
	tab rating

* MODERATORS

	* respondent's education 
	tab xtess191
	recode xtess191 (1/2=1 "high ed") (3/4=0 "low ed"), gen(resp_educ)
	tab resp_educ

********************************************************************************
* ANALYSIS

*Test-Hyp1: Within concerted cultivation vignette, rating for parenting style will be higher among high-education vs. low-education respondents.

	reg rating i.resp_educ if conc_cult==1
	// reject. 0.079
	tess 1.resp_educ +, init(Ishizuka661)
	
*Test-Hyp2: Within concerted cultivation vignette, rating for parenting style will be lower for mothers vs. fathers.
	reg rating i.parent_mother if conc_cult==1
	// reject. 0.460
	tess 1.parent_mother -
	
*Test-Hyp3: Within natural growth vignette, rating for parenting style will be lower for mothers vs. fathers.
	reg rating i.parent_mother if conc_cult==0
	// reject. 0.200
	tess 1.parent_mother -
	
*Test-Hyp4: Within low-education respondents, rating will be higher for concerted cultivation vs. natural growth parenting.
	reg rating i.conc_cult if resp_educ==0
	// do not reject. 0.000	
	tess 1.conc_cult +
	
*Test-Hyp5: Within high-education respondents, rating will be higher for concerted cultivation vs. natural growth parenting.
	reg rating i.conc_cult if resp_educ==1
	// do not reject. 0.000	
	tess 1.conc_cult +	
